


Première partie

Énoncés

1. Courbes


1.1. Cercle trigonométrique

- Les vecteurs unitaires font 25 mm, $x \in [-1.2, 1.2]$, et, $y \in [-1.2, 1.95]$;
- le cercle est le cercle unité « `unitsize` », l'angle θ fait 60° .




1.2. Arctangente

- Les vecteurs unitaires font 15 mm, $x \in [-5, 5]$, et, $y \in [-1.8, 2.5]$;
- le nom de la fonction Arctangente est « `atan` ».




1.3. Parabole

- Les vecteurs unitaires font 12 mm ;
- le paramètre p de la parabole, $y^2 = 2px$, est 3 : on utilisera le paramétrage suivant :
 - $x = \frac{t^2}{2p}$;
 - $y = t$;
- $t \in [-2.3p, 2.3p]$ $x \in [-p, 3p]$, et, $y \in [-2.3p, 2.3p]$;
- le point M ici choisi correspond à $t = 1.8p$.




1.4. Newton-Ramphson

- la figure fait 16 cm de haut, le repère est orthonormal ;
- $x \in [-1, 5.5]$, et $y \in [-3.5, 3.5]$;
- la figure sera coupée aux dimensions annoncées ;
- la fonction est $x \rightarrow 3 - \frac{(x - 4.3)^2}{2}$;
- sa dérivée est $x \rightarrow -(x - 4.3)$;
- $a = 0.5$, $b = 4$ et $x_n = 3.3$;
- une fois la valeur de x_n choisie, le reste est automatique !



1.5. Intégrale double


- Les vecteurs unitaires font 22 mm ;
- $x \in [-0.5, 6]$, et, $y \in [-0.5, 5]$;
- La courbe est décrite en paramétriques par les deux fonctions :
 - $x = 3 + 1.5 \cos(t) + \sin(t)$;
 - $y = 2.4 + 1.8 \sin(t)$;
- On utilisera « `dir time` » pour les tangentes verticales ;
- La valeur de x où on coupe est 2.8 ;
- On utilisera « `intersectionpoint` » pour la construction qui en dépend, et qui, ensuite, est automatique...



1.6. Ellipse optimale

Il s'agit de l'ellipse, centrée et d'aire minimale, contenant les 4 points P_1 à P_4 , de coordonnées respectives $(3, 0)$, $(3, \sqrt{3})$, $(0, \frac{1}{2})$ et $(3, \frac{1}{2})$.

- Les vecteurs unitaires font 17 mm ;
- $x \in [-4.5, 4.5]$, et $y \in [-2.5, 2.5]$;
- La courbe est décrite en paramétriques par les deux fonctions :
 - $x = 3 \cos(t) + 3 \sin(t)$;
 - $y = \sqrt{3} \sin(t)$;




2. Géométrie plane élémentaire

2.1. Rotation


- Prendre un triangle quelconque ;
- tournez le de 60° en marquant un angle, autour d'un point Ω choisi ;
- symétriser le triangle de départ par rapport à Ω .

On visualise assez bien que la symétrie centrale est une rotation de 180° !



2.2. Sur le triangle rectangle

- Les vecteurs unitaires font 14 mm ;
- on prend un triangle rectangle d’hypoténuse horizontale de 4 unités et d’angles 60° et 30° ;
- on abaisse la hauteur et on découpe en 2 triangles qu’on translate ;
- on symétrise par rapport à des droites, de façon que les hypoténuses deviennent horizontales ;
 - ces droites sont donc parallèles aux bissectrices des angles concernés ;
- on obtient deux triangles homothétiques de (ABC) .



Ceci constitue, bien sûr, une démonstration élémentaire du théorème de Pythagore !

Deuxième partie

Corrigés

3. Courbes

3.1. Cercle trigonométrique

```

import graph;

// Dimension des vecteurs unitaires
unitsize(25mm);

// Variables
real xmin=-1.2,xmax=1.2,ymin=-1.2,ymax=1.95;

// Le cercle
draw(unitcircle);
draw(arc((0,0),.3,0,60),red,ArcArrow(SimpleHead));
label(scale(.7)*"$\theta$",(.3,.2));
draw((1,ymin)--(1,ymax),red);
draw((0,0)--(1.2,1.2*sqrt(3)),blue);
draw((cos(pi/3),0)--(cos(pi/3),sin(pi/3))--(0,sin(pi/3)),dashed+blue);
draw((1,tan(pi/3))--(0,tan(pi/3)),dashed+blue);
label(scale(.7)*"$\cos(\theta)$",(cos(pi/3),0),S);
label(scale(.7)*"$\sin(\theta)$",(0,sin(pi/3)),W);
label(scale(.7)*"$\tan(\theta)$",(0,tan(pi/3)),W);

// Tracé des axes
xaxis(Label("$x$",position=EndPoint,align=NE),
      xmin=xmin,xmax=xmax,
      Arrow);

yaxis(Label("$y$",position=EndPoint,align=E),
      ymin=ymin,ymax=ymax,
      Arrow);

label("$0$",(0,0),SE);
draw((0,0)--(1,0),Arrow);
draw((0,0)--(0,1),Arrow);

ylimits(ymin,ymax,Crop);

```

3.2. Arctangente

```

import graph;

// Dimension des vecteurs unitaires
unitsize(15mm);

// Variables
real xmin=-5,xmax=5,ymin=-1.8,ymax=2.2;

// Tracé du graphe
real f(real t) {return atan(t);}
path Gf=graph(f,xmin,xmax,operator ..);
draw(Gf,blue);
draw((xmin,pi/2)--(xmax,pi/2),red);
draw((xmin,-pi/2)--(xmax,-pi/2),red);

label("$y=\mathrm{arctan}(x)$",(-1.5,f(-1.5)),NW);

// Tracé des axes
xaxis(Label("$x$",position=EndPoint,align=NE),
      xmin=xmin,xmax=xmax,
      Arrow);

yaxis(Label("$y$",position=EndPoint,align=E),
      ymin=ymin,ymax=ymax,
      Arrow);

label("$0$",(0,0),SE);
draw((0,0)--(1,0),Arrow);
draw((0,0)--(0,1),Arrow);

draw((-1,pi/2)--(1,pi/2));label(scale(.7)*"$\pi/2$",(0,pi/2),NE);
draw((-1,-pi/2)--(1,-pi/2));label(scale(.7)*"$-\pi/2$",(0,-pi/2),NE);

draw((1,0)--(1,f(1))--(0,f(1)),dashed+blue);label(scale(.7)*"$\pi/4$",(0,f(1)),W);

ylimits(ymin,ymax,Crop);

```

3.3. Parabole

```

import graph;
// Dimension des vecteurs unitaires
unitsize(12mm);
// Variables
real p=3;
real tmin=-2.3*p,tmax=2.3*p;
real xmin=-p,xmax=3*p;
real ymin=tmin,ymax=tmax;
real c=p/2,d=-p/2;
// Définition de la fonction et du point M
real f(real t) {return t^2/(2*p);}
real g(real t) {return t;}
real t0=1.8*p;
pair M=(f(t0),g(t0));
// Tracé de la courbe
path Gf=graph(f,g,tmin,tmax,operator ..);
draw(Gf,linewidth(.7bp)+red);
label("\fbox{\mathscr{P}}\;:\;y^2=2p\,x$", (1.5*p,p),E);
// Tracé des axes
xaxis(Label("$x$",position=EndPoint,align=NE),
      xmin=xmin,xmax=xmax,
      Arrow);
yaxis(Label("$y$",position=EndPoint,align=E),
      ymin=ymin,ymax=ymax,
      Arrow);
label("$0$", (0,0),SW);
draw((0,0)--(1,0),Arrow);
draw((0,0)--(0,1),Arrow);
// Sommets, foyer
dot((c,0),blue);label("$F$", (c,0),N);label("$c$", (c,0),S);
// Directrice
dot((d,0),blue);
label("$d$", (d,0),SE);
label("$D$", (d,ymax),SW);
draw((d,ymin)--(d,ymax),blue);
// Tracés relatifs au point M
draw((d+.3,0)--(d+.3,.3)--(d,.3),blue);
draw((c,0)--M--(d,M.y),blue);
dot(M,blue);label("$M$", M,N);
dot((d,M.y),blue);label("$H$", (d,M.y),W);
draw((d+.3,M.y)--(d+.3,M.y+0.3)--(d,M.y+0.3),blue);
// Théorie
label("$c=\dfrac{p}{2}=OF$", (1.5*p,-.8),E);
label("$e=1=\dfrac{MF}{MH}$", (1.5*p,-1.8),E);
label("$d=-\dfrac{p}{2}$", (1.5*p,-2.8),E);

```

3.4. Newton-Ramphson

```

import graph;
// Dimension des vecteurs unitaires
size(0,16cm);
// Variables
real xmin=-1,xmax=5.5,ymin=-3.5,ymax=3.5;
real a=0.5,b=4;
real xn=3.3;
// Tracé du graphe
real f(real t) {return 3-((t-4.3)^2)/2;}
path Gf=graph(f,xmin,xmax,operator ..);
draw(Gf,blue);
// La dérivée
real ff(real t) {return 4.3-t;}

// Le premier point et sa tangente
dot((xn,f(xn)),red);
draw((xn,f(xn))--(xn,0),dashed);
label("$x_n$",(xn,0),S);
real ftxn(real t) {return f(xn)+(t-xn)*ff(xn);}
path Gftxn=graph(ftxn,a,b,operator ..);
draw(Gftxn,red);

// Le point d'intersection, le point suivant et sa tangente
real xnn=xn-f(xn)/ff(xn);
dot((xnn,f(xnn)),red);
draw((xnn,f(xnn))--(xnn,0),dashed);
label("$x_{n+1}$",(xnn,0),SE);
real ftxnn(real t) {return f(xnn)+(t-xnn)*ff(xnn);}
path Gftxnn=graph(ftxnn,a,b,operator ..);
draw(Gftxnn,red);

// Le point suivant
real xnnn=xnn-f(xnn)/ff(xnn);
dot((xnnn,f(xnnn)),red);
draw((xnnn,f(xnnn))--(xnnn,0),dashed);

// Tracé des axes
xaxis(Label("$x$",position=EndPoint,align=NE),
      xmin=xmin,xmax=xmax, Arrow);
yaxis(Label("$y$",position=EndPoint,align=E),
      ymin=ymin,ymax=ymax,Arrow);

label("$0$",(0,0),SW);
draw((.5,.05)--(.5,-.05));label("$a$",(.5,-.05),S);
draw((4,.05)--(4,-.05));label("$b$",(4,-.05),S);

ylimits(ymin,ymax,Crop);

```

3.5. Intégrale double

```

import graph;
// Dimension des vecteurs unitaires
unitsize(22mm);
// Variables
real xmin=-.5,xmax=6,ymin=-.5,ymax=5;
xlimits(xmin,xmax,Crop);
ylimits(ymin,ymax,Crop);

real x(real t) {return 3+1.5*cos(t)+sin(t);}
real y(real t) {return 2.4+1.8*sin(t);}

path C1=graph(x,y,0,pi);
path C2=graph(x,y,pi,2*pi);
draw(C1,heavygreen);
draw(C2,heavygreen);

real t0=1.5;
label("$\Delta$",(x(t0),y(t0)),N);

real ta=dirtime(C2,(0,-1));
pair pA=point(C2,ta);
draw((pA.x,0)--pA+(0,1),dashed+blue);label("$a\phantom{b}$",(pA.x,0),S,blue);
dot(pA,blue);
real tb=dirtime(C1,(0,1));
pair pB=point(C1,tb);
draw((pB.x,0)--pB+(0,1),dashed+blue);label("$b$",(pB.x,0),S,blue);
dot(pB,blue);

real x0=2.8;
path segmentx0=(x0,0)--(x0,ymax-.5);
draw(segmentx0,blue);label("$x\phantom{b}$",(x0,0),S,blue);

pair pC=intersectionpoint(segmentx0,C2);
draw((0,pC.y)--pC,dashed+red);label("$u(x)$",(0,pC.y),W,red);
dot(pC,red);
pair pD=intersectionpoint(segmentx0,C1);
draw((0,pD.y)--pD,dashed+red);label("$v(x)$",(0,pD.y),W,red);
dot(pD,red);

// Tracé des axes
xaxis(Label("$x$",position=EndPoint,align=NE),
      xmin=xmin,xmax=xmax,Arrow);
yaxis(Label("$y$",position=EndPoint,align=E),
      ymin=ymin,ymax=ymax,Arrow);

label("$0$",(0,0),SW);

```

3.6. Ellipse optimale

```

unitsize(17mm);
import graph;

import patterns;

real x(real t) {return 3*cos(t)+3*sin(t);}
real y(real t) {return sqrt(3)*sin(t);}
draw((-4.5,0)--(4.5,0),Arrow(HookHead));label("$x$", (3.8,0),N);
draw((0,0)--(1,0),Arrow(HookHead));label("$\overrightarrow{j}$", (0,1),E);
draw((0,-2.5)--(0,2.5),Arrow(HookHead));label("$y$", (0,2.4),E);
draw((0,0)--(0,1),Arrow(HookHead));label("$\overrightarrow{i}$", (1,0),S);

path zonehachuree=buildcycle(graph(x,y,0,pi),graph(x,y,pi,2*pi));
add("hachure",hatch(H=7mm,dir=NE,lightred));
fill(zonehachuree,pattern("hachure"));

draw(graph(x,y,0,2*pi),blue);
draw((3,0)--(3,sqrt(3)),green);
dot((3,0),red);label("$P_1$", (3,0),S);
dot((3,sqrt(3)),red);label("$P_2$", (3,sqrt(3)),N);
dot((0,0.5),green);label("$P_3$", (0,0.5),E);
dot((3,0.5),green);label("$P_4$", (3,0.5),E);

```

4. Géométrie plane élémentaire

4.1. Rotation

```

import geometry;
size(12cm,0);

pair pA=(0,0),pB=(3,0),pC=(2,1);
path TriBase=pA--pB--pC--cycle;
draw(TriBase,.8bp+green);
label("$A$",pA,NW);
label("$B$",pB,NE);
label("$C$",pC,N);

pair Omega=(-.5,-1);
transform r=rotate(60,Omega);
dot("$\Omega$",Omega,SE);

draw(r*TriBase,.8bp+red);
draw(pA--Omega--r*pA,dashed+.25pt+red);
draw(pB--Omega--r*pB,dashed+.25pt+blue);
draw(pC--Omega--r*pC,dashed+.25pt+green);
path angle=arc(pA,Omega,r*pA,.7);
draw(scale(.7)*"60",angle,heavyred,Arrow);

draw(shift(Omega)*scale(-1)*shift(-Omega)*TriBase,.8bp+blue);
draw(Omega--shift(Omega)*scale(-1)*shift(-Omega)*pA,dashed+.25pt+red);
draw(Omega--shift(Omega)*scale(-1)*shift(-Omega)*pB,dashed+.25pt+blue);
draw(Omega--shift(Omega)*scale(-1)*shift(-Omega)*pC,dashed+.25pt+green);

```

4.2. Sur le triangle rectangle

```

import geometry;
unitsize(14mm);
pair pA=(0,0),pB=(4,0),pC=(1,sqrt(3));
path TriBase=pA--pB--pC--cycle;
draw(TriBase);
label("$A$",pA,NW);
label("$B$",pB,NE);
label("$C$",pC,N);
pair pH=(1,0);
draw(pC--pH);
label("$H$",pH,S);

path TriGauche=pA--pH--pC--cycle;
draw(shift(-2,-3)*TriGauche,blue);
label("$A_1$",shift(-2,-3)*pA,NW);
label("$C_1$",shift(-2,-3)*pC,N);
label("$H_1$",shift(-2,-3)*pH,E);
draw(pH--shift(-2,-3)*pH,dashed+green,Arrow);
path TriDroit=pH--pB--pC--cycle;
draw(shift(4,-3)*TriDroit,red);
label("$C_2$",shift(4,-3)*pC,NW);
label("$B_2$",shift(4,-3)*pB,NE);
label("$H_2$",shift(4,-3)*pH,W);
draw(pH--shift(4,-3)*pH,dashed+green,Arrow);

pair pE=(-1,-4),dpE=(sqrt(3),1);
draw(pE-dpE--pE+1.5*dpE,blue);
draw(reflect(line(pE,pE+dpE))*shift(-2,-3)*TriGauche,blue);
label("$A'_1$",reflect(line(pE,pE+dpE))*shift(-2,-3)*pA,NW);
label("$C'_1$",reflect(line(pE,pE+dpE))*shift(-2,-3)*pC,NE);
label("$H'_1$",reflect(line(pE,pE+dpE))*shift(-2,-3)*pH,NE);
draw(shift(-2,-3)*pH--reflect(line(pE,pE+dpE))*shift(-2,-3)*pH,dashed+green,Arrow);
pair pF=(4,-3.193),dpF=(1,-tan(pi/12));
draw(pF-dpF--pF+5*dpF,red);
draw(reflect(line(pF,pF+dpF))*shift(4,-3)*TriDroit,red);
label("$C'_2$",reflect(line(pF,pF+dpF))*shift(4,-3)*pC,NW);
label("$B'_2$",reflect(line(pF,pF+dpF))*shift(4,-3)*pB,NE);
label("$H'_2$",reflect(line(pF,pF+dpF))*shift(4,-3)*pH,W);
draw(shift(4,-3)*pH--reflect(line(pF,pF+dpF))*shift(4,-3)*pH,dashed+green,Arrow);

draw(shift(0,-8)*TriBase,magenta);
label("$A''$",shift(0,-8)*pA,SW);
label("$B''$",shift(0,-8)*pB,SE);
label("$C''$",shift(0,-8)*pC,N);

draw(reflect(line(pE,pE+dpE))*shift(-2,-3)*pA--shift(0,-8)*pA,dashed+green,Arrow);
draw(reflect(line(pF,pF+dpF))*shift(4,-3)*pB--shift(0,-8)*pB,dashed+green,Arrow);

```
